Computation of Curve Staking out Coordinates on the Excel Spreadsheet

T.G.Gacoki

Kenya National Highways Authority, Kenha

SUMMARY

A procedure for the computation of curve setting out data on the Excel spreadsheet is outlined. An Excel worksheet with five interrelated sheets computes the curve setting out data. In sheet one the included angles are computed from the coordinates entries of the intersection points of the straights. In sheet two curve elements for the various curves are computed from the entries of the radii and lengths of the transition curves and the included angles computed in sheet one. Sheet three computes the chainages of the principal points of the curves and the coordinates of these points. Sheet four and sheet five computes the coordinates of the center- line of the points along the straights and curves respectively.

Computation of Curve Staking out Coordinates on the Excel Spreadsheet

T.G.Gacoki

Kenya National Highways Authority, Kenha

1. INTRODUCTION

When working in an engineering environment, the surveyor is called upon time and again to set out engineering works. This is the case in road construction. Curve setting out may be necessary either, during the actual road construction or for determining the extent of road reserve for purposes of land acquisition where necessary. The need may also become apparent for purposes of showing any encroachments on the road reserve. In the latter case, where cadastral information is in cassini coordinates, while the engineering data is in UTM coordinates, conversion between the two systems would be necessary. This conversion was demonstrated in **[1]** on the Excel spreadsheet.

The design engineer provides setting out data, usually being generated via engineering software. The data is accompanied by a design drawing showing the various elements of the design data. However, where the print out of the staking out coordinates for the center line is unavailable and the surveyor has the possession of the design drawing showing the various elements of the curves, staking out data can be conveniently computed by use of the Excel spreadsheet.

Presented here are steps and procedures necessary for the computation of the staking out coordinates of the centerline of the curve data. The steps are simple since no prior knowledge of programming is necessary as the functions used are in-built in the Excel spreadsheet. Furthermore, the computation of curve setting coordinates is repetitive, and can therefore be conveniently carried out on the Excel spreadsheet. Also presented is the computation of running chainages through a road project.

2. ELEMENTS OF THE CIRCULAR CURVE

The horizontal alignment of a road consists of a series of straights and curves. Where two straights meet, they create an intersection point. The designed road consists of sections of straights and curves between straights for the road under design. The curves are defined by the deflection angles (difference in bearing) between straights and the radius of the curve. It therefore follows that the initial requirements for setting out any curve are the location of the straights and their intersection points. The intersection angles are either provided by the design engineer or through direct field measurements. Once the intersection angles are provided together with the radius the curve can be set out.

3. ELEMENTS OF THE TRANSITION CURVE

A transition curve is one in which the curvature varies uniformly with respect to arc, in order to allow a gradual change from one radius to another (a straight being a curve of infinite radius) to permit a gradual change in the super elevation. It must of course have the same radius of curvature at its ends, as the circular curve it links. The transition curve must have a constant rate of change of curvature with respect to arc. The solution presented here assumes a curve with two equal transitions and a circular curve in the middle.

The various elements of the circular and transition curves are shown in appendix 2, with the formulae for the various elements are as shown in **[2]** and the figure 2 is as shown in **[3].**

4. THE SOLUTION

As an example, we start with seven intersection points with known coordinates are used to compute the six included angles formed from the seven intersection points. There after, the various elements (radii, transition curve length) are entered to compute curve parameters like the shifts (for transition curves), tangent lengths, curve lengths etc. Subsequently, the coordinates of the principal points and running chainages are computed.

The solution is illustrated by the use of five different sheets on the same worksheet. The first three sheets compute the various elements including the coordinates of the principal points. Thereafter, the fourth worksheet is used to compute the coordinates of the straights, while the fifth worksheet computes the coordinates of the two spirals and the circular curve.

Sheet 1: Computation of intersection angles

The coordinates of the intersection points are entered in this sheet and are used to calculate the intersection angles for the curves formed by these points. The coordinates of the intersection points (IPs) are entered in cell C6:D13 for 7 intersection points as shown in Table 1. This is used as an example; otherwise the number of points to be entered is not limited. From these entries the distances and the bearings of the straights, followed by the intersection angles are computed in columns E, F and G respectively as follows:

Distances

The distance between the intersection points is computed from the following: -

E7:= ((C7-C6)^2+(D7-D6)^2)^0.5

This formulae is then copied to cells E8:E13

Bearing

The bearings between the straights are computed from the following:-

`F7:=IF(ATAN2((C7-C6),(D7-D6))<0,(ATAN2((C7-C6),(D7-D6))+2*PI())*180/PI(),(ATAN2((C7-C6),(D7-D6))*180/PI()))`

Table 1: Computation of total deflection angles from the intersection points

	A	B	C	D	E	F	G	H	I	J	K	L
1												
2	**njoro turnoff to timboroa - calculation of total deflection angles**											
3												
4					**UNIT**	**METRES**						
5	**intersection points**	**curve no.**	**NORTHINGS**	**EASTINGS**	**dist**	**brg**	**delta (total deflection angle)**	**deg**	**min**	**sec**	**delta (total deflection angle)**	**CURVE "R" OR "L"**
6	IP0		9968890.58	841709.08								
7	IP1	1	9968182.71	839814.42	2022.573	249.5135816	41.01887	41	1	7.94	41.018872	R
8	IP2	2	9968503.43	838958.09	914.421	290.5324531	4.03722	4	2	13.99	4.0372196	R
9	IP3	3	9969988.31	835710.29	3571.145	294.5696728	25.27551	25	16	31.83	-25.27551	L
10	IP4	4	9969980.21	835052.89	657.449	269.2941653	40.74782	40	44	52.15	40.747819	R
11	IP5	5	9970420.47	834529.00	684.315	310.0419838	15.12237	15	7	20.52	15.122366	R
12	IP6	6	9971521.77	833762.56	1341.758	325.1643499	38.24795	38	14	52.62	-38.24795	L
13	IP7		9972883.70	829284.54	4680.544	286.9163989						

Sheet 1

The copy command is used to copy this formula to cells F8:F12. The explanation for the derivation of this formula is shown in appendix 1. In cell G7, the total deflection angles are computed by entering the following formula.

G7: =IF(ABS(F7-F8)>180,360-ABS(F7-F8),ABS(F7-F8))

This is then copied to cells F8:F12. The angles obtained are the total deflection angles irrespective of direction (either right or left); Columns H, I, and J computes the degrees, minutes and seconds of the included angle as follows: -

H7: = INT(G7)
I7: =INT((G7-H7)*60)
J7: =((G7-H7)*60-I7)*60

The formulae are the copied to cells H8: J12
Lastly, the total deflection angles are computed showing the direction (either left or right). This is accomplished by entering the following formula:-

K7:= IF(F8-F7>180,(F8-F7)-360,IF(F8-F7<-180,(F8-F7)+360,F8-F7))

This is then copied to cells K8:K12. A positive angle means the curve formed from the intersection of the two straights is a right hand curve and vice versa.

Sheet 2: Computation of Curve Elements

The total deflection angles computed in Table1 are used to compute the coordinates of the principal points of the curves. Prior to this however, certain curve elements need to be computed. These are the shifts, transition deflection angles, deflection angles for the circular curve, and cartesian coordinates for the transition curves. Other necessary elements include the total deflection distance between the origin and the end of the transition curve, the tangent distances for the curves, length of the circular curves and the distances of straights between the curves.

For the computations to be possible the radii of the various curves and the lengths of the transition curves must be provided. For the purpose of this paper it will be assumed that the transition curves are identical on both sides of the circular curve. The computation of these elements are illustrated in Table 2

Table 2: Computation of curve elements

	A	B	C	D	E	F	G	H	I	J	K	L	M
1													
2	njoro turnoff to timboroa - calculation of curve elements												
3	unit metres												
4													
5	**curve no.**	**curve "R" or "L"**	**Radius**	**Lt**	**s (shifts)**	**transition deflection angle**	**deflection angle for circular curve**	**Total X**	**Total Y**	**total deflection distance for the transition curve L**	**Total tangent length T**	**circular curve length Lc**	**Distances of straights between curves**
6											0		
7	1	R	467.322	0	0	0	41.018872	0	0	0	174.81	334.56	1847.761
8	2	R	4973.02	0	0	0	4.037220	0	0	0	175.28	350.41	564.330
9	3	L	870	80	0.30651	2.63428871	20.006930	79.983	1.226	79.992	235.14	303.79	3160.730
10	4	R	870	80	0.30651	2.63428871	35.479241	79.983	1.226	79.992	363.21	538.73	59.103
11	5	R	870	80	0.30651	2.63428871	9.853789	79.983	1.226	79.992	155.52	149.62	165.584
12	6	L	870	70	0.23467	2.30500262	33.637946	69.989	0.939	69.995	336.75	510.77	849.486

Sheet 2

The formulae necessary for the computation of these elements are shown in appendix 2 and are computed as follows:-

Shift
E7:= D7^2/(24*C7)
This formula is copied to cells E8:E12

Transition deflection angle
F7:= (D7/(2*C7))*180/PI()
This formula is copied to cells F8:F12

Deflection angle for the circular curve
G7: =Sheet1!G7-2*F7
This formula is copied to cells G8:G12

Cartesian coordinates
H7: =D7-(D7^3/(40*C7^2))

I7: =(D7^2)/(6*C7)-(D7^4)/(336*C7^3)
This formula is copied to cells H8:H12 and I8:I12 respectively
Total deflection distance for the transition curve
J7:= IF(D7=0,"0",D7-(D7^5/(90*(D7*C7)^2))+(D7^9/(22680*(D7*C7)^4)))
This formula is copied to cells J8:J12

Tangent length
K7: =H7-C7*SIN(F7*PI()/180)+((C7+E7)*TAN((Sheet1!G7/2)*PI()/180))
This formula is copied to cells K8:K12

Length of circular curve
L7: =C7*((Sheet1!G7-2*F7)*PI()/180)
This formula is copied to cells M8:M12

Length of straights between curves
M7: =Sheet1!E7-(K6+K7)
This formula is copied to cells N8:N12

Sheet 3: Computing coordinates of the principal points of the curves

The curve elements computed in Table 2 are used to compute the running chainages of the principal points and their coordinates. Eventually, these coordinates of the principal points are used to compute the setting out data for the horizontal alignment. The principal points are as follows:

Table 3: Computing coordinates and chainages of the principal curve points

	A	B	C	D	E	F	G	H	I	J	K	L	M	N
1														
2	njoro turnoff to timboroa - calculation of chainages and coordinates of the principal points													
3														
4	unit metres													
5	curve no.	curve "R" or "L"	chainage TS	COORDINATES TS		chainage SC/BCC	COORDINATES SC/BCC		chainage CS/ECC	COORDINATES CS/ECC		chainage CT	COORDINATES CT	
6				Northings	Eastings		Northings	Eastings		Northings	Eastings		Northings	Eastings
7														
8			8657.240									8657.240	9968890.580	841709.079
9	1	R	10505.001	9968243.891	839978.179	10505.001	9968243.891	839978.179	10839.564	9968244.022	839650.716	10839.564	9968244.022	839650.716
10	2	R	11403.893	9968441.954	839122.236	11403.893	9968441.954	839122.236	11754.306	9968576.312	838798.684	11754.306	9968576.312	838798.684
11	3	L	14915.036	9969890.542	835924.138	14995.036	9969922.684	835850.887	15298.828	9969985.174	835555.166	15378.828	9969985.414	835475.174
12	4	R	15437.931	9969984.686	835416.076	15517.931	9969984.927	835336.084	16056.661	9970163.364	834836.851	16136.661	9970213.883	834774.829
13	5	R	16302.244	9970320.411	834648.063	16382.244	9970372.806	834587.618	16531.868	9970483.166	834486.857	16611.868	9970548.116	834440.163
14	6	L	17461.353	9971245.368	833954.916	17531.353	9971302.278	833914.166	18042.124	9971598.496	833507.062	18112.124	9971619.759	833440.374

Sheet 3

TS	The tangent to the transition curve (transition curve origin).
SC/BCC	The beginning of the circular curve
CS/ECC	The end of the circular curve (circular curve to transition).
CT	Transition to tangent

The running chainages and the coordinates of these principal points are computed as is illustrated in Table 3. The starting chainage is entered in cell L8 and the corresponding coordinates in cell M8:N8

TS

Chainage: C9:=L8+Sheet2!M7
Northings: D9:= Sheet1!C7+Sheet2!K7*COS((Sheet1!F7-180)*PI()/180)
Eastings: E9:= Sheet1!D7+Sheet2!K7*SIN((Sheet1!F7-180)*PI()/180)

These formulae are copied to cells C10:E14

SC/BCC

Chainage: F9:=C8+Sheet2!D7
Northings: G9:=IF(B9="R",D9+Sheet2!J7*COS((Sheet1!F7+(Sheet2!F7/3))*PI()/180), D9+Sheet2!J7*COS((Sheet1!F7-(Sheet2!F7/3))*PI()/180))
Eastings:
H9:=IF(B9="R",E9+Sheet2!J7*SIN((Sheet1!F7+(Sheet2!F7/3))*PI()/180), E9+Sheet2!J7*SIN((Sheet1!F7-(Sheet2!F7/3))*PI()/180))

These formulae are copied to cells F10:H14

CS/ECC

Chainage: I9:=F9+Sheet2!L7
Northings:
J9:=IF(B9="R",G9+(Sheet2!C7*2*SIN((Sheet2!G7/2)*PI()/180))*COS((Sheet1!F7+Sheet2!F7+Sheet2!G7/2)*PI()/180),G9+(Sheet2!C7*2*SIN((Sheet2!G7/2)*PI()/180))*COS((Sheet1!F7-Sheet2!F7- Sheet2!G7/2) *PI()/180))*COS((Sheet1!F7-Sheet2!F7-Sheet2!G7/2)*PI()/180))
Eastings:
K9:=IF(B9="R",H9+(Sheet2!C7*2*SIN((Sheet2!G7/2)*PI()/180))*SIN((Sheet1!F7+Sheet2!F7+Sheet2!G7/2)*PI()/180), H9+(Sheet2!C7*2*SIN((Sheet2!G7/2)*PI()/180))*SIN((Sheet1!F7-Sheet2!F7- Sheet2!G7/2)*PI()/180))

These formulae are copied to cells I10:K14

CT

Chainage: L9:=L8+Sheet2!M7+2*Sheet2!D7+Sheet2!L7
Northings: M9:=Sheet1!C7+Sheet2!K7*COS(Sheet1!F8*PI()/180)
Eastings: N9:= Sheet1!D7+Sheet2!K7*SIN(Sheet1!F8*PI()/180)

These formulae sre copied to cells L10:N14

Sheet 4: Computation of center-line coordinates for the straights

This is a straightforward computation as only the bearing of the straight is used to compute the coordinates of the preceding chainages. However, the starting chainage, bearing and coordinates for the particular straight are copied from sheet 1 and 3.table 4. As an example we will compute the coordinates of the centerline of the straights between curves 4 and 5

The starting chainage for the particular straight in this case given by: -

```
B6:=Sheet3!L12
```

The subsequent chainages are obtained by inserting the following formula:-

```
B7:= IF(B6+((INT(B6/20))+1-(B6/20))*20<$B$6+Sheet2!$M$11,B6+((INT(B6/20))+1-
(B6/20))*20,IF(B6=$B$6+Sheet2!$M$11,"",IF(B6+((INT(B6/20))+1-
(B6/20))*20>$B$6+Sheet2!$M$11,$B$6+Sheet2!$M$11,"")))
```

This formula returns a null value “” or *#value* when the length of straight is exceeded, acting as a check.

The bearing for the particular straight is inserted from sheet 1as follows: -

```
G6=Sheet1!$F$11
```

Since this bearing is constant it is simply copied to the other cells i.e G7:G15. The coordinates of the starting chainage are copied from sheet 3 as follows:-

Northings: I6=Sheet3!M12
Eastings: J6=Sheet3!N12

Finally, the coordinates of the chainages of the straights are computed as follows:-

Northings: I7:= I6+C7*COS(G7*PI()/180)
Eastings: J7:= J6+C7*SIN(G7*PI()/180)

These formulae are then copied to cells I7:J15.

Table 4: Computation of center-line coordinates for the straights

	A	B	C	D	E	F	G	H	I	J
1										
2	**njoro turnoff to timboroa- computation of centre-line coordinates for the straights**									
3										
4		**chainage**			**deflection**	**distance**	**deflection brg**	**tangential. Brg**	**Northings**	**Eastings**
5										
6	CT4	**16136.661**	**0**				**310.0419838**	**310.0419838**	**9970213.883**	**834774.829**
7		16140	3.3393849				310.0419838	310.0419838	9970216.031	834772.272
8		16160	20				310.0419838	310.0419838	9970228.898	834756.961
9		16180	20				310.0419838	310.0419838	9970241.765	834741.650
10		16200	20				310.0419838	310.0419838	9970254.632	834726.338
11		16220	20				310.0419838	310.0419838	9970267.499	834711.027
12		16240	20				310.0419838	310.0419838	9970280.366	834695.715
13		16260	20				310.0419838	310.0419838	9970293.233	834680.404
14		16280	20				310.0419838	310.0419838	9970306.100	834665.092
15		16300	20				310.0419838	310.0419838	9970318.967	834649.781
16	TS5	16302.244	2.2442698				310.0419838	310.0419838	9970320.411	834648.063
17			#VALUE!						#VALUE!	#VALUE!
18		#VALUE!	#VALUE!						#VALUE!	#VALUE!

Straight 5: Straight between curves 4 and 5

Sheet 5: Computation of chainages and staking out coordinates for the first transition curve

The computation of the staking out coordinates of the first transition curve starts from the last chainage of the preceding straight. The computation is illustrated in table 5 and is computed as follows:-

Table 5: Computation of center-line coordinates for the first transition curve

	A	B	C	D	E	F	G	H	I	J
1										
2	**njoro turnoff to timboroa- computation of centre-line coordinates for the first transition curve**									
3										
4		**chainage**			**deflection angle from TS5**	**distance from TS5**	**deflection brg from TS5**	**tangential. Brg**	**Northings**	**Eastings**
5										
6					**69600**					
7	**TS5**	**16302.2443**	**0**	**0**	**0**	**0**	**310.04198377**	**310.04198377**	**9970320.411**	**834648.063**
8		16320.000	17.756	17.756	0.04325529	17.756	310.085239061	310.17174964	9970331.844	834634.478
9		16340.000	20	37.756	0.19558138	37.756	310.237565154	310.62872843	9970344.799	834619.241
10		16360.000	20	57.756	0.45766762	57.754	310.499651396	311.41499332	9970357.919	834604.146
11		16380.000	20	77.756	0.82950692	77.749	310.871490695	312.53054429	9970371.287	834589.270
12	**SC/BCC5**	16382.244	2.244	80	0.87808052	79.992	310.920064295	312.67627249	9970372.806	834587.618
13			#VALUE!	#VALUE!	#VALUE!	#VALUE!	#VALUE!	#VALUE!	#VALUE!	#VALUE!
14		#VALUE!	#VALUE!	#VALUE!	#VALUE!	#VALUE!	#VALUE!	#VALUE!	#VALUE!	#VALUE!

Curve 5: first transition curve

The northings (N) and easting (E) of the TS point are known (having already been computed in Table 3). In addition from the equations in appendix 2, various points on the transition curve can be computed. Using these points we could extract (N,E) values of various chainages of the spiral. The Northings and Eastings values are computed at 20m intervals along the spiral as shown in Table 5.

The starting chainage for the TS in this case, TS5 given by:-

B7:=Sheet3!C13

The subsequent chainages are obtained by inserting the formula:

B8:= IF(B7+((INT(B6/20))+1-(B7/20))*20<$B$7+Sheet2!$D$11,B7+((INT(B7/20))+1-(B7/20))*20,IF(B7=$B$7+Sheet2!$D$11,"",IF(B7+((INT(B7/20))+1-(B7/20))*20>$B$7+Sheet2!$D$11,$B$7+Sheet2!$D$11,"")))

The formula returns a null value “ ” or *#value* when the length of the transition curve is exceeded acting as a check.

The deflection angle i.e. the angle subtended at TS5 by the extended tangent and the chord connecting TS and any arbitrary point on the spiral (positive if angle is right hand and negative if left hand) is computed from the following:-

E7:=((D7^2/(6*$E$6))-((0.0762*(D7^2/(6*$E$6))^3)+(0.0166*(D7^2/(6*$E$6))^5)))*180/PI()

The length of the spiral chord from TS5 to any given point on the spiral is computed from:-
F7:= IF(D7=0,"0",D7-(D7^5/(90*($E$6)^2))+(D7^9/(22680*($E$6)^4)))

The deflection bearing is given by

Bearing from preceding straight: G7:=Sheet1!F11
Deflection bearing: G8:= IF(Sheet1!$L$11="R",$G$7+E8,$G$7-E8)

The tangential bearing is given by

Bearing from preceding straight: H7:=Sheet1!F11
Tangential bearing: H8:= IF(Sheet1!$L$11="R", $H$7+((D8^2/(2*$E$6))*180/PI()),$H$7-((D8^2/(2*$E$6))*180/PI()))

The coordinates of TS5 are copied from sheet 3 as follows:-

Northings: I7:=Sheet1!D13
Eastings: J7:= Sheet1!E13

Finally the coordinates of a point on the spiral are given by

Northings: I8:= $I$7+(F8*COS(G8*PI()/180))
Eastings: J8:=$J$7+(F8*SIN(G8*PI()/180))

Sheet 5: Computation of chainages and staking out coordinates for the circular curve

Table 6: Computation of center-line coordinates for the circular curve

	L	M	N	O	P	Q	R	S	T	U
1										
2	**njoro turnoff to timboroa- computation of centre-line coordinates for the curve - circular**									
3										
4		**chainage**			**deflection angle from SC5**	**distance from SC5**	**deflection brg from SC5**	**tangential. Brg**	**Northings**	**Eastings**
5										
6					**870**					
7	**SC/BCC5**	**16382.244**	**0**	**0**	**0.000000**	**0.000**	**312.67627249**	**312.67627249**	**9970372.806**	**834587.618**
8		16400.000	17.756	17.756	0.584671	17.755	313.26094398	313.84561548	9970384.975	834574.688
9		16420.000	20.000	37.756	1.243244	37.753	313.91951616	315.16275984	9970398.994	834560.424
10		16440.000	20.000	57.756	1.901816	57.745	314.57808834	316.47990419	9970413.337	834546.487
11		16460.000	20.000	77.756	2.560388	77.730	315.23666052	317.79704855	9970427.996	834532.883
12		16480.000	20.000	97.756	3.218960	97.704	315.89523270	319.11419291	9970442.965	834519.619
13		16500.000	20.000	117.756	3.877532	117.666	316.55380488	320.43133726	9970458.234	834506.703
14		16520.000	20.000	137.756	4.536105	137.612	317.21237705	321.74848162	9970473.797	834494.141
15	**CS/ECC5**	16531.868	11.868	149.624	4.926894	149.439	317.60316682	322.53006114	9970483.166	834486.857
16			#VALUE!	#VALUE!	#VALUE!	#VALUE!	#VALUE!	#VALUE!	#VALUE!	#VALUE!
17		#VALUE!	#VALUE!	#VALUE!	#VALUE!	#VALUE!	#VALUE!	#VALUE!	#VALUE!	#VALUE!

Curve 5: Circular Curve

The starting chainage for the particular circular curve in this case given by:-
M7:=Sheet3!F13

The subsequent chainages are obtained by inserting the formula:

M8:= IF(M7+((INT(M7/20))+1-(M7/20))*20<$M$7+Sheet2!$L$11,M7+((INT(M7/20))+1-(M7/20))*20,IF(M7=$M$7+Sheet2!$L$11,"",IF(M7+((INT(M7/20))+1-(M7/20))*20>$M$7+Sheet2!$L$11,$M$7+Sheet2!$L$11,"")))

The formula returns a null value “ ” or *#value* when the length of the circular curve is exceeded acting as a check.

The deflection angle from the beginning of the circular curve (SC/BCC5)

P7:= ((O7/$P$6)*180/PI())/2

The distance from the beginning of the circular curve to a point on the circular curve

Q7:= 2*$P$6*SIN(P7*PI()/180)

The deflection bearing is given

Start bearing: R7:= IF(Sheet2!B11="R",Sheet1!F11+Sheet2!F11,Sheet1!F11-Sheet2!F11)

Deflection bearing: R8:= IF(Sheet1!$L$11="R",$R$7+P8,$R$7-P8)

The tangential bearing is given by

Start bearing: S7:= IF(Sheet2!B11="R",Sheet1!F11+Sheet2!F11,Sheet1!F11-Sheet2!F11)

Tangential bearing: S8:= IF(Sheet1!$L$11="R",$S$7+2*P8,$S$7-2*P8)

The coordinates of TS5 are copied from sheet 3 as follows:-

Northings: T7:= Sheet3!G13
Eastings: U7:= Sheet3!H13

Finally the coordinates of a point on the circular curve are given by

Northings: T8:= $T$7+(Q8*COS(R8*PI()/180))
Eastings: U8:= $U$7+(Q8*SIN(R8*PI()/180))

This formula is copied to cells T9:U12

Sheet 5: Computation of chainages and staking out coordinates for the second transition curve

The starting chainage for the particular second transition curve in this case given by:-

X7:=Sheet3!I13

Table 7: Computation of center-line coordinates for the second transition curve

	W	X	Y	Z	AA	AB	AC	AD	AE	AF	AG	AH
1												
2	**njoro turnoff to timboroa- computation of centre-line coordinates for the curve - transition**											
3												
4		**chainage**			**deflection from CT5**	**distance from CT5**	**phi**	**distance from CS5**	**deflection bearing from CS5**	**tangential. Brg**	**Northings**	**Eastings**
5					**69600**							
6												
7	**CS/ECC5**	**16531.868**	**0.000**	**80**	**0.8780805**	**79.992**	**2.634288713**	**0.000**	**322.530061143**	**322.530061143**	**9970483.166**	**834486.857**
8		16540.000	8.132	71.868	0.7086400	71.863	2.125944670	8.132	322.788769979	323.038405186	9970489.643	834481.939
9		16560.000	20.000	51.868	0.3691104	51.867	1.107334567	28.131	323.347831372	324.057015290	9970505.735	834470.064
10		16580.000	20.000	31.868	0.1393368	31.868	0.418010552	48.129	323.797133433	324.746339304	9970522.003	834458.430
11		16600.000	20.000	11.868	0.0193242	11.868	0.057972627	68.126	324.136676983	325.106377229	9970538.377	834446.945
12	**CT5**	16611.868	11.868	0.000	0.0000000	0	0.000000000	79.992	324.286269335	325.164349856	9970548.116	834440.163
13			#VALUE!	#VALUE!	#VALUE!	#VALUE!	#VALUE!	#VALUE!	#VALUE!	#VALUE!	#VALUE!	#VALUE!
14		#VALUE!	#VALUE!	#VALUE!	#VALUE!	#VALUE!	#VALUE!	#VALUE!	#VALUE!	#VALUE!	#VALUE!	#VALUE!

Curve 5: Second transition curve

The subsequent chainages are obtained by inserting the formula:

X8:= IF(X7+((INT(X7/20))+1-(X7/20))*20<$X$7+Sheet2!$D$11,X7+((INT(X7/20))+1-(X7/20))*20,IF(X7=$X$7+Sheet2!$D$11,"",IF(X7+((INT(X7/20))+1-(X7/20))*20>$X$7+Sheet2!$D$11,$X$7+Sheet2!$D$11,"")))

The formula returns a null value “ ” or *#value* when the length of the second transition curve is exceeded acting as a check.

The deflection angle from CT5- end of transition curve

AA7:= ((Z7^2/(6*$AA$5))-((0.0762*(Z7^2/(6*$AA$5))^3)+(0.0166*(Z7^2/(6*$AA$5))^5)))*180/PI()

Distance from CT5

AB7:= IF(Z7=0,"0",Z7-(Z7^5/(90*($AA$5)^2))+(Z7^9/(22680*($AA$5)^4)))

Phi: AC7:= (Z7^2/(2*$AA$5))*180/PI()

Distance from CS5- end of circular curve

AD7:= ($AB$7^2+AB7^2-2*$AB$7*AB7*COS(($AA$7-AA7)*PI()/180))^0.5

The deflection bearing is given

Start bearing: AE7:= IF(Sheet2!B11="R",Sheet1!F11+Sheet2!F11+Sheet2!G11,Sheet1!F11-Sheet2!F11-Sheet2!G11)

Deflection bearing: AE8:= IF(Sheet2!$B$11="R",$AE$7+(($AC$7-$AA$7)-(ACOS((AD8^2+$AB$7^2-AB8^2)/(2*AD8*$AB$7)))*180/PI()),$AE$7-(($AC$7-$AA$7)-(ACOS((AD8^2+$AB$7^2-AB8^2)/(2*AD8*$AB$7)))*180/PI()))

The tangential bearing is given by

Start bearing: AF7:= IF(Sheet2!B11="R",Sheet1!F11+Sheet2!F11+Sheet2!G11,Sheet1!F11-Sheet2!F11-Sheet2!G11)

Tangential bearing: AF8:= IF(Sheet1!$L$11="R", $AE$7+($AC$7-AC7),$AE$7-($AC$7-AC8))

The coordinates of TS5 are copied from sheet 3 as follows:-

Northings: AG7:= Sheet3!J13
Eastings: AH7:= Sheet3!K13

Finally the coordinates of a point on the spiral are given by

Northings: AG8:= $AG$7+AD8*COS(AE8*PI()/180)
Eastings: AH8:= $AH$7+AD8*SIN(AE8*PI()/180)

5. CONCLUSION

The results obtained are the same as can be obtained from a computer program. This use of the Excel Spreadsheet can be used to illustrate the computation of curve setting out data. The final computed coordinates can be converted to text format from Excel and transferred to a GIS software (Arc GIS for example) and be used to show the road centerline. Within the GIS software buffers can be applied for the road reserve, which in turn can be used for land acquisition where this is necessary. Indeed, the same can be used to determine properties encroaching on the road reserve.

The results shown here were taken from a project which had been successfully used in the construction of a road project and the results obtained were similar to those generated from a computer program.

6. APPENDIX 1

Computation of bearings between intersection points
Given the coordinates of two intersection points (IP0,IP1) as shown on figure 1, the bearing, α between the two points can be computed from the following:-

$$\alpha = 2 \times ARCTAN\left(\frac{\Delta E}{S + \Delta N}\right)$$

Where
ΔE is the difference in Eastings between the two points
ΔN is the difference in Northings between the two points
S is the distance between the two points

This formula can be illustrated by use of half angle formulae, however, the same can be shown graphically as follows:-



FIGURE 1

The line from IP0 is prolonged to *C* by distance *S*. Thereafter *C* is then joined to IP1 as shown in figure 1 above. It can be clearly seen from figure 1 that the angle formed at *C* is *α/2*. From figure 1,

$$\tan\frac{\alpha}{2} = \frac{\Delta E}{S + \Delta N}$$

$$\frac{\alpha}{2} = ARCTAN\left(\frac{\Delta E}{S + \Delta N}\right)$$

$$\alpha = 2 * ARCTAN\left(\frac{\Delta E}{S + \Delta N}\right)$$

In excel this is entered as

α=ATAN2(ΔN,ΔE)

The function ATAN2 operates on ΔN and ΔE to give angles in radians in the range -π to π, excluding -π. If both ΔN and ΔE are zero, Excel returns the message #DIV/0 or division by zero. Positive angles are in the first and second quadrant and are correct whole circle bearings. Negative angles are in the third and fourth quadrant and 2*π is added to convert to whole circle bearings.

APPENDIX 2

Elements of Transition and Circular Curves



Figure 2: *Elements of Transition and Circular Curves*

Following are the key parameters that explain this geometry:-

Lt	Length of spiral from TS to SC- *see table 2*
IP	Point of horizontal intersection point (not shown on figure 2)
TS/TC	Point where spiral begins (Tangent to Spiral- spiral origin)
SC/BCC	Point where spiral ends and circular curve starts (Beginning of circle from spiral end)
φ	Spiral angle (or) deflection angle between tangent and tangential direction at end of spiral
CS/ECC	End of circle to spiral
CT	Spiral to Tangent (second spiral – not shown in figure 2)
K	TA Abscissa of the shifted curve PC referred to TS (or tangent) distance at shifted PC from TS
s	Shift of circular curve- *see table 2*

O	Center point of circular curve
Δc	Angle subtended by circular curve
Δ	Total deflection angle between the two tangents (2φ+Δc)
R	Radius of circular curve - *see table 2*
δ	Deflection angle for the circular curve for a chord of length *a* is given by $\frac{a}{2R}$ in radians
Lc	circular curve length $2R\Delta$c -*see table 2*
Tc	Tangent length of circular curve $R\tan\frac{\Delta c}{2}$
T	Total (extended) tangent length from TS to IP given by

$T=(R+s)\tan\frac{\Delta}{2}+K$ - *see table 2*

Where $K=TotalX-R\sin\varphi$

X Total X tangent distance at SC from TS

$TotalX=L-\frac{L^3}{40R^2}+\frac{L^5}{3456R^4}-...$ as shown in **[2]** , L is the full length of the transition curve.- *see table 2*

Y Total Y = D2D off set distance at SC from (Tangent at) TS

$TotalY=\frac{L^2}{6R}-\frac{L^4}{336R^3}+\frac{L^6}{42240R^5}-...$ as shown in **[2]** , L is the full length of transition curve.

s AB The offset of initial tangent into the PC of shifted curve (shift of the circular curve).

$s=AB=AE-BE=TotalY-(R-R\cos\Delta\varphi)=\frac{L^3}{24LR}-\frac{L^7}{2668(LR)^3}$ as shown in **[2]**.

δ Deflection angle from TS for the spiral given by

$\frac{\varphi}{3}-\left\{0.0762\left(\frac{\varphi}{3}\right)^3+0.0166\left(\frac{\varphi}{5}\right)^5+...\right\}$ where $\varphi=\frac{c^2}{2lR}$ as shown in **[2]**.

l Chord length from TS is given by $l = c - \frac{c^5}{90(lR)^2} + \frac{c^9}{22680\,(lR)^4}$ for the spiral as shown in **[2]**, where *c* is distance from the beginning of the spiral.

REFERENCES

Gacoki, T.G., 2006. Conversion of Cassini Coordinates to UTM on Excel spreadsheet. Survey review, Vol. 38. No. 302. 689-696

Allan, A.L., Howlley, J.R. and Maynes, J.H.B., 1977. Practical Field Surveying and Computations. Heinemann. 689 pages.

http://www.eldoradosoft.com/curves.htm